

## 9/11.2

# Le traitement de texte Weka

---

Nous avons vu comment créer un traitement de texte élémentaire en turbo Pascal dans la Partie 4, chapitre 4.5.4. Ce traitement de texte aurait pu être amélioré dans de larges mesures si la mémoire des CPC était plus étendue. Comme ce n'est pas le cas, nous allons voir comment créer un traitement de texte pleine page totalement écrit en Assembleur.

### 9/11.2.1

## Fonctions élémentaires

---


### I - Saisie, affichage et mémorisation de caractères sur l'écran en pleine page

Le programme est implanté en &8000. Il débute par une série d'initialisations :

- Passage en mode 2 grâce à la macro **SCR SET MODE** du Firmware ;
- Mémorisation de l'adresse d'implantation du début du texte dans la variable **DEBTEX** ;
- Initialisation de la zone de stockage du texte à 32D (caractère espace).

Ces initialisations sont suivies d'une boucle d'attente principale dans laquelle le programme acquiert les caractères tapés au clavier. S'il s'agit de caractères affichables, il les affiche à la position courante du curseur et les mémorise. S'il s'agit de caractères de contrôle valides, il active le sous-programme de traitement correspondant. S'il s'agit de l'appui simultané sur **Ctrl Q**, le programme est avorté.

La logique apparaît clairement dans l'ordinogramme suivant :




Examinons en détails les diverses actions effectuées par le programme.

- Lorsque le caractère tapé a un code ASCII supérieur ou égal à 32, il est affiché à la position courante du curseur.
- L'appui sur la touche <Enter> provoque un passage à la ligne suivante, éventuellement suivi d'un scrolling vers le haut de l'écran.
- L'appui sur une des touches flèche provoque le déplacement contrôlé du curseur dans la direction voulue, éventuellement suivi d'un scrolling vers le haut ou vers le bas.
- L'appui simultané sur **Ctrl 2** et **Z** provoque un scrolling de l'écran vers le haut.
- L'appui simultané sur **Ctrl** et **W** provoque un scrolling de l'écran vers le bas.
- La touche <Del> permet d'effacer le caractère à gauche du curseur et déplace les caractères qui suivent d'une position vers la gauche.
- La touche <Clr> permet d'effacer le caractère courant et déplace les caractères qui suivent d'une position vers la gauche.

### Affichage d'un caractère

Tout caractère de code ASCII supérieur ou égal à 32 est affiché et mémorisé selon la logique de l'ordinogramme suivant :




Le sous-programme chargé de cette tâche a pour nom AFFICH.

### Affichage d'une ligne

L'affichage d'une ligne consiste :

- au positionnement du curseur sur l'écran,
- en l'affichage de 80 caractères consécutifs à partir de la position initiale du curseur.

Ces actions s'enchaînent comme suit :



Le sous-programme chargé de cette tâche a pour nom WRILIG.


### Appui sur la touche <Enter>

Lorsque l'utilisateur appuie sur la touche <Enter>, le caractère courant (qui est affiché en inverse vidéo) est réaffiché avec un attribut normal. L'adresse courante du début de la nouvelle ligne est calculée.

Si le curseur se trouve sur la ligne 25, un scrolling haut est effectué et le curseur est déplacé au début de la nouvelle ligne 25.


Dans le cas contraire, le curseur est simplement déplacé au début de la ligne suivante.

Ces actions s'enchaînent comme suit :



Le sous-programme chargé de cette tâche a pour nom TCR.

TCR fait appel à un sous-programme de restitution de caractère qui permet d'afficher le caractère courant avec un attribut d'affichage normal. Ce sous-programme a pour nom RESTIT. La logique qu'il met en œuvre est la suivante :



## II - Gestion du curseur (haut, bas, droit et gauche)


### Appui sur UP

Lorsque l'utilisateur appuie sur la touche flèche **UP**, le caractère courant est restitué avec un attribut normal (pas inverse vidéo).

Si le curseur était positionné sur la première ligne de texte, aucune action n'est effectuée.

Dans le cas contraire, un scrolling d'une ligne vers le bas est effectué si nécessaire et la nouvelle position du curseur est mémorisée.

Ces actions s'enchaînent comme suit :



Le sous-programme chargé de cette tâche a pour nom TUP.


### Appui sur DOWN

Lorsque l'utilisateur appuie sur la touche flèche **DOWN**, le caractère courant est restitué avec un attribut normal (pas inverse vidéo).

Si le curseur était positionné sur la dernière ligne de l'écran, un scrolling d'une ligne vers le haut est effectué, et la nouvelle ligne 25 est affichée.

Si la ligne courante n'est pas la 25<sup>e</sup>, le curseur est simplement déplacé d'une position vers le bas.

Ces actions s'enchaînent comme suit :



Le sous-programme chargé de cette tâche a pour nom TDOWN.




**Appui sur LEFT**

Lorsque la touche flèche **LEFT** est pressée, le curseur se déplace d'une position vers la gauche dans le cas où sa position colonne courante est différente de 1.

Dans le cas contraire, aucune action n'est effectuée.

Ces actions s'enchaînent comme suit :




Le sous-programme chargé de cette tâche a pour nom TLEFT.

### Appui sur RIGHT

Lorsque la touche flèche **RIGHT** est pressée, le curseur se déplace d'une position vers la droite dans le cas où sa position colonne courante est différente de 80.

Dans le cas contraire, aucune action n'est effectuée.

Ces actions s'enchaînent comme suit :



Le sous-programme chargé de cette tâche a pour nom TRIGHT.


### **III - Commandes de scrolling bas et haut d'une ligne d'écran**

#### **Scrolling bas**

Lorsque les touches **Ctrl** et **W** sont pressées simultanément, l'écran tout entier est déplacé vers le bas d'une ligne (si cela est possible). Pour ce faire, le programme calcule l'adresse du premier caractère de la ligne supérieure de l'écran. Si cette adresse est celle du début du texte, le scrolling ne peut être effectué.

Dans le cas contraire, le caractère courant est restitué (attribut d'affichage normal), un scrolling vers le bas est effectué et une nouvelle première ligne est affichée.

Ces actions s'enchaînent comme suit :




Le sous-programme chargé de cette tâche a pour nom BSCROL.

### Scrolling haut

Un scrolling vers le haut est toujours possible. Il s'effectue lorsque l'utilisateur appuie simultanément sur les touches **Ctrl** et **Z**. Le caractère courant est restitué avec un attribut d'affichage normal, l'adresse de début de 25<sup>e</sup> ligne est calculée et la 25<sup>e</sup> ligne affichée.

Ces actions s'enchaînent comme suit :




Le sous-programme chargé de cette tâche a pour nom HSCROL.

## IV - Touches CLR et DEL pour l'effacement d'un caractère

### Appui sur la touche <Clr>

Lorsque l'utilisateur appuie sur la touche <Clr>, le caractère courant est effacé et les caractères qui suivent sont déplacés d'une position vers la gauche. Pour ce faire, la puissante instruction LDI est utilisée pour déplacer un bloc mémoire et la ligne est réaffichée.

Ces actions s'enchaînent comme suit :




Le sous-programme chargé de cette tâche a pour nom TCLR.

**Appui sur la touche <Del>**

Lorsque l'utilisateur appuie sur la touche <Del>, le caractère courant et les caractères qui suivent sont déplacés d'une position vers la gauche. Pour ce faire, la puissante instruction LDI est également utilisée pour déplacer un bloc mémoire et la ligne modifiée est réaffichée.

Ces actions s'enchaînent comme suit :



Le sous-programme chargé de cette tâche a pour nom TDEL.

Le sous-programme Assembleur a été écrit sous ZEN. En voici le listing :

```

1          ORG 8000H
2          LOAD 8000H
3          ;=====
4          ; Initialisation
5          ;=====
6 8000 3E02          LD  A,2
7 8002 CD0EBC       CALL MODE          ;Passage en MODE 2
8 8005 210090       LD  HL,DEBTEX          ;Debut du texte
9 8008 22B0B2       LD  (ADRLIB),HL      ;à ligne depart
10         ;
11 800B 210090       LD  HL,9000H          ;Init memoire travail
12 800E 010010       LD  BC,1000H
13 8011 1620        LD  D,32
14         BOUINI:  EQU $                ;Boucle d'init
15 8013 72          LD  (HL),D
16 8014 0B         DEC  BC
17 8015 23         INC  HL
18 8016 78         LD  A,B
19 8017 B1         OR   C
20 8018 20F9       JR   NZ,BOUINI
21         ;
22         ;=====
23         ;Acquisition d'un caractere
24         ;et action correspondante
25         ;=====
26         ACQCAR:  EQU $
27 801A CDBABB     CALL PLCUR          ;Affiche curseur
28 801D CD06BB     CALL WAIT           ;Attente frappe 1 caract
29 8020 FE0D       CP   CR
30 8022 CA55B0     JP   Z,TCR          ;Traitement CR

```



```

31 8025 FEF0          CP   UP
32 8027 CAB80        JP   Z,TUP          ;Traitement UP
33 802A FEF1          CP   DOWN
34 802C CADE80       JP   Z,TDOWN        ;Traitement DOWN
35 802F FEF2          CP   LEFT
36 8031 CA1181       JP   Z,TLEFT        ;Traitement LEFT
37 8034 FEF3          CP   RIGHT
38 8036 CA2881       JP   Z,TRIGHT       ;Traitement RIGHT
39 8039 FE1A          CP   CTRLZ
40 803B CA9D81       JP   Z,HSCROL       ;Scrolling haut
41 803E FE17          CP   CTRLW
42 8040 CA3781       JP   Z,BSCROL       ;Scrolling bas
43 8043 FE10          CP   CLR
44 8045 CAE881       JP   Z,TCLR          ;Traitement CLR
45 8048 FE7F          CP   DEL
46 804A CA0582       JP   Z,TDEL         ;Effact 1 caract
47 804D FE11          CP   CTRLQ
48 804F CB           RET   Z          ;Fin du programme
49 8050 C32F82       JP   AFFICH        ;Affiche un caract
50 8053 18C5         JR   ACQCAR        ;Boucle principale

51          ;=====
52          ; ZONE DES TRAITEMENTS
53          ;=====
54          ;
55          ;-----
56          ; Traitement du CR
57          ;-----

58          TCR:      EQU   $
59 8055 CD7082       CALL RESTIT
60 805B 2A8082       LD   HL,(ADRLIG)

```

```

61 805B 115000      LD   DE,80
62 805E 19          ADD  HL,DE
63 805F 228082     LD   (ADRLIG),HL      ;Nile à courante
64 8062 CD78BB     CALL GETCUR
65 8065 7D         LD   A,L
66 8066 FE19       CP   25              ;Si ligne 25
67 8068 280C       JR   Z,TCR1          ;Trait special
68 806A CD78BB     CALL GETCUR
69 806D 2601       LD   H,1
70 806F 2C         INC  L
71 8070 CD75BB     CALL SETCUR
72 8073 C31A80     JP   ACQCAR
73                TCR1:      EQU  $
74 8076 0601       LD   B,1
75 8078 AF         XOR  A
76 8079 CD4DBC     CALL HWROLL          ;Scrolling haut
77 807C CD5282     CALL WRILIG         ;Affich nile ligne
78 807F CD78BB     CALL GETCUR
79 8082 2601       LD   H,1
80 8084 CD75BB     CALL SETCUR         ;Posit. curseur
81 8087 C31A80     JP   ACQCAR
82                ;
83                ;-----
84                ; Traitement du UP
85                ;-----
86                TUP:      EQU  $
87 808A CD7082     CALL RESTIT         ;Restit car courant
88 808D CD78BB     CALL GETCUR
89 8090 7D         LD   A,L              ;Colonne
90 8091 FE01       CP   1

```

```

91 8093 2816          JR   Z,TUP1
92 8095 2A80B2        LD   HL,(ADRLIG)
93 8098 115000        LD   DE,80
94 809B ED52          SBC  HL,DE
95 809D 2280B2        LD   (ADRLIG),HL
96 80A0 CD78BB        CALL GETCUR
97 80A3 2D            DEC  L
98 80A4 25            DEC  H
99 80A5 CD75BB        CALL SETCUR
100 80A8 C31A80       JP   ACQCAR
101                   TUP1: EQU  $
102 80AB 210090       LD   HL,DEBTEX           ;Debut du texte
103 80AE ED5B80B2    LD   DE,(ADRLIG)
104 80B2 ED52          SBC  HL,DE
105 80B4 200A         JR   NZ,TUP2
106 80B6 CD78BB        CALL GETCUR
107 80B9 25            DEC  H
108 80BA CD75BB        CALL SETCUR
109 80BD C31A80       JP   ACQCAR
110                   TUP2: EQU  $
111 80C0 0600          LD   B,0
112 80C2 AF            XOR  A
113 80C3 CD4DBC        CALL HWROLL           ;Scroll
114 80C6 2A80B2        LD   HL,(ADRLIG)
115 80C9 115000        LD   DE,80
116 80CC ED52          SBC  HL,DE
117 80CE 2280B2        LD   (ADRLIG),HL     ;Nlle à debut ligne
118 80D1 CD52B2        CALL WRILIG
119 80D4 CD78BB        CALL GETCUR
120 80D7 25            DEC  H
    
```

```

121 80D8 CD75BB          CALL SETCUR
122 80DB C31A80          JP   ACQCAR

123                      ;
124                      ;-----
125                      ; Traitement du DOWN
126                      ;-----
127          TDOWN:      EQU  $
128 80DE CD7082          CALL RESTIT          ;Restit car courant
129 80E1 2A8082          LD   HL,(ADRLIG)
130 80E4 115000          LD   DE,80
131 80E7 19              ADD  HL,DE
132 80E8 228082          LD   (ADRLIG),HL
133 80EB CD78BB          CALL GETCUR
134 80EE 7D              LD   A,L          ;Colonne
135 80EF FE19            CP   25
136 80F1 280B           JR   Z,TD01
137 80F3 CD78BB          CALL GETCUR
138 80F6 25             DEC  H
139 80F7 2C             INC  L
140 80F8 CD75BB          CALL SETCUR
141 80FB C31A80          JP   ACQCAR
142          TD01:      EQU  $
143 80FE 0601           LD   B,1
144 8100 AF             XOR  A
145 8101 CD4DBC          CALL HWROLL
146 8104 CD5282          CALL WRILIG
147 8107 CD78BB          CALL GETCUR
148 810A 25             DEC  H
149 810B CD75BB          CALL SETCUR
150 810E C31A80          JP   ACQCAR

```

```

151          ;
152          ;-----
153          ; Traitement du LEFT
154          ;-----
155          TLEFT:      EQU  $
156 8111 CD78BB          CALL GETCUR
157 8114 7C              LD  A,H
158 8115 FE01            CP   1
159 8117 CA1A80          JP   Z,ACQCAR
160 811A CD7082          CALL RESTIT          ;Restit caract courant
161 811D CD78BB          CALL GETCUR
162 8120 25              DEC  H
163 8121 25              DEC  H
164 8122 CD75BB          CALL SETCUR
165 8125 C31A80          JP   ACQCAR
166          ;
167          TRIGHT:     EQU  $
168 8128 CD78BB          CALL GETCUR
169 812B 7C              LD  A,H
170 812C FE50            CP   80
171 812E CA1A80          JP   Z,ACQCAR
172 8131 CD7082          CALL RESTIT          ;Restit car courant
173 8134 C31A80          JP   ACQCAR
174          ;
175          ;-----
176          ; Scrolling bas
177          ;-----
178          BSCROL:      EQU  $
179 8137 CD78BB          CALL GETCUR
180 813A 228782          LD  (SAVCUR),HL
    
```

```

181 813D 45          LD  B,L
182 813E 2A8082     LD  HL,(ADRLIG)
183 8141 115000     LD  DE,80
184 8144 37         SCF
185 8145 3F         CCF
186                BSC1: EQU  $
187 8146 ED52       SBC  HL,DE
188 8148 10FC       DJNZ BSC1
189 814A 19         ADD  HL,DE
190 814B 228382     LD  (SAUVHL),HL      ;à 1e ligne
191 814E 110090     LD  DE,DEBTEX
192 8151 ED52       SBC  HL,DE
193 8153 7C         LD  A,H
194 8154 B5         OR   L
195 8155 CA1A80     JP   Z,ACQCAR      ;Scrolling impossible
196 8158 CD78BB     CALL GETCUR
197 815B 0600       LD  B,0
198 815D 4C         LD  C,H
199 815E 2A8082     LD  HL,(ADRLIG)
200 8161 09         ADD  HL,BC
201 8162 2B         DEC  HL
202 8163 7E         LD  A,(HL)
203 8164 CD5ABB     CALL WRCHAR      ;Ecr caract lu
204 8167 0600       LD  B,0
205 8169 AF         XOR  A
206 816A CD4DBC     CALL HWROLL      ;Scrolling
207 816D 210101     LD  HL,101H
208 8170 CD75BB     CALL SETCUR
209 8173 2A8082     LD  HL,(ADRLIG)
210 8176 228582     LD  (SAUV2),HL    ;Sauv à ligne

```

```

211 8179 2A8382      LD   HL, (SAUVHL)
212 817C 115000      LD   DE, 80
213 817F ED52        SBC  HL, DE
214 8181 228082      LD   (ADRLIG), HL
215 8184 CD5282      CALL WRILIG
216 8187 2A8782      LD   HL, (SAVCUR)
217 818A CD75BB      CALL SETCUR
218 818D 2A8582      LD   HL, (SAUV2)
219 8190 115000      LD   DE, 80
220 8193 37          SCF
221 8194 3F          CCF
222 8195 ED52        SBC  HL, DE
223 8197 228082      LD   (ADRLIG), HL
224 819A C31A80      JP   ACQCAR

225                  ;
226                  ;-----
227                  ; Scrolling haut
228                  ;-----
229                  HSCROL:   EQU  $
230 819D CD78BB      CALL GETCUR
231 81A0 228782      LD   (SAVCUR), HL      ;Sauv à curseur
232 81A3 0600        LD   B, 0
233 81A5 4C          LD   C, H
234 81A6 2A8082      LD   HL, (ADRLIG)
235 81A9 09          ADD  HL, BC
236 81AA 2B          DEC  HL
237 81AB 7E          LD   A, (HL)
238 81AC CD5ABB      CALL WRCHAR      ;Restit car au curseur
239 81AF 0601        LD   B, 1
240 81B1 AF          XOR  A
    
```

```

241 81B2 CD4DBC          CALL HWROLL
242 81B5 211901         LD   HL,119H
243 81B8 CD75BB          CALL SETCUR
244 81BB 2A8082         LD   HL,(ADRLIG)
245 81BE 228582         LD   (SAUV2),HL      ;Sauv à ligne
246 81C1 ED5B8782       LD   DE,(SAVCUR)
247 81C5 3E1A           LD   A,26
248 81C7 93             SUB  E
249 81C8 47             LD   B,A
250 81C9 115000         LD   DE,80
251                    HSC1: EQU  $
252 81CC 19             ADD  HL,DE
253 81CD 10FD           DJNZ HSC1
254 81CF 228082         LD   (ADRLIG),HL
255 81D2 CD5282         CALL WRILIG          ;Aff dernière ligne
256 81D5 2A8582         LD   HL,(SAUV2)
257 81D8 115000         LD   DE,80
258 81DB 19             ADD  HL,DE
259 81DC 228082         LD   (ADRLIG),HL    ;Restit à ligne
260 81DF 2A8782         LD   HL,(SAVCUR)
261 81E2 CD75BB          CALL SETCUR          ;Restit à curseur
262 81E5 C31A80         JP   ACQCAR
263                    ;
264                    ;-----
265                    ; Traitement du CLR
266                    ;-----
267                    TCLR: EQU  $
268 81E8 CD78BB          CALL GETCUR
269 81EB 3E50           LD   A,80
270 81ED 94             SUB  H

```



```

271 81EE 47          LD   B,A
272 81EF 04          INC  B
273 81F0 1600        LD   D,0
274 81F2 5C          LD   E,H
275 81F3 2A80B2      LD   HL,(ADRLIG)
276 81F6 19          ADD  HL,DE
277 81F7 2B          DEC  HL
278 81F8 54          LD   D,H
279 81F9 5D          LD   E,L
280 81FA 23          INC  HL
281                 TCLR1: EQU  $
282 81FB EDA0        LDI
283 81FD 10FC        DJNZ TCLR1
284 81FF CD52B2      CALL WRILIG           ;Aff nle ligne
285 8202 C31A80      JP   ACQCAR
286                 ;
287                 ;-----
288                 § Traitement du DEL
289                 ;-----
290                 TDEL:   EQU  $
291 8205 CD78BB      CALL GETCUR
292 8208 7C          LD   A,H
293 8209 FE01        CP   1
294 820B CA1A80      JP   Z,ACQCAR
295 820E 3E50        LD   A,80
296 8210 94          SUB  H
297 8211 47          LD   B,A
298 8212 04          INC  B
299 8213 1600        LD   D,0
300 8215 5C          LD   E,H

```

```

301 8216 2A80B2      LD   HL, (ADRLIG)
302 8219 19          ADD  HL, DE
303 821A 2B          DEC  HL
304 821B 54          LD   D, H
305 821C 5D          LD   E, L
306 821D 1B          DEC  DE
307                  TDEL1: EQU  *
308 821E EDA0        LDI
309 8220 10FC        DJNZ TDEL1
310 8222 CD52B2      CALL WRILIG
311 8225 CD78BB      CALL GETCUR
312 8228 25          DEC  H
313 8229 CD75BB      CALL SETCUR
314 822C C31A80      JP   ACQCAR
315                  ;
316                  ;-----
317                  ; Affichage d'un caractere
318                  ;-----
319                  AFFICH: EQU  *
320 822F FE20        CP   32
321 8231 DA1A80      JP   C, ACQCAR          ;Caract non affichable
322 8234 328282      LD   (SAUVA), A          ; Sauv caract tape
323 8237 CD78BB      CALL GETCUR          ;Position curseur
324 823A 4C          LD   C, H          ;Colonne curseur
325 823B 7C          LD   A, H
326 823C FE50        CP   80
327 823E CA1A80      JP   Z, ACQCAR
328 8241 0600        LD   B, 0
329 8243 2A80B2      LD   HL, (ADRLIG)      ;Adresse ligne
330 8246 09          ADD  HL, BC

```

```

331 8247 2B          DEC  HL
332 8248 3A8282     LD   A, (SAUVA)
333 824B 77         LD   (HL),A           ;Sauvegarde caract
334 824C CD5ABB     CALL WRCHAR         ;Affiche caract
335 824F C31A80     JP   ACQCAR
336                ;
337                ;-----
338                ; Affichage d'une ligne
339                ;-----
340                WRILIG: EQU  $
341 8252 CD78BB     CALL GETCUR
342 8255 228382     LD   (SAUVHL),HL
343 8258 2601       LD   H,1
344 825A CD75BB     CALL SETCUR
345 825D 0650       LD   B,80
346 825F 2A8082     LD   HL, (ADRLIG)
347                BOUWRI: EQU  $
348 8262 7E         LD   A, (HL)
349 8263 CD5ABB     CALL WRCHAR
350 8266 23         INC  HL
351 8267 10F9       DJNZ BOUWRI
352 8269 2A8382     LD   HL, (SAUVHL)
353 826C CD75BB     CALL SETCUR
354 826F C9         RET
355                ;
356                ;-----
357                ;Restitution du caractere courant
358                ;-----
359                RESTIT: EQU  $
360 8270 CD78BB     CALL GETCUR

```

```

361 8273 4C          LD   C,H
362 8274 0600        LD   B,0
363 8276 2A80B2      LD   HL,(ADRLIG)
364 8279 09          ADD  HL,BC
365 827A 2B          DEC  HL
366 827B 7E          LD   A,(HL)
367 827C CD5ABB      CALL WRCHAR
368 827F C9          RET

369                ;
370                ;=====
371                ;   ZONE DES EQU
372                ;=====
373                DEBTX:   EQU  9000H           ;Debut du texte
374                WAIT:   EQU  0BB06H          ;KM WAIT CHAR
375                WRCHAR: EQU  0BB5AH          ;TXT OUTPUT
376                MODE:   EQU  0BC0EH          ;SCR SET MODE
377                GETCUR: EQU  0BB78H          ;TXT GET CURSOR
378                PLCUR:  EQU  0BB8AH          ;PLACE CURSOR
379                SETCUR: EQU  0BB75H          ;TXT SET CURSOR
380                HWROLL: EQU  0BC4DH          ;SCR HW ROLL
381                CURDIS: EQU  0BB7EH          ;CUR DISABLE
382                CURON:  EQU  0BB81H          ;TXT CUR ON
383                CR:      EQU  13              ;Carriage Return
384                UP:      EQU  240
385                DOWN:    EQU  241
386                LEFT:    EQU  242
387                RIGHT:   EQU  243
388                CTRLA:   EQU  01H
389                CTRLF:   EQU  06H
390                CTRLZ:   EQU  1AH

```

```

391          CTRLW:      EQU  17H
392          CTRLC:      EQU  03H
393          CTRLR:      EQU  12H
394          CLR:         EQU  16
395          DEL:        EQU  127
396          CTRLT:      EQU  14H
397          CTRLY:      EQU  19H
398          CTRLV:      EQU  16H
399          CTRLQ:      EQU  11H
400          SPACE:     EQU  32
401          ;
402          ;=====
403          ; ZONE DES BUFFERS
404          ;=====
405          ADRLIG:     DS   2           ;à debut de ligne
406          SAUVA:     DS   1           ;Sauvegarde registre A
407          SAUVHL:    DS   2           ;Sauvegarde reg HL
408          SAUV2:     DS   2           ;Sauvegarde 2 octets
409          SAVCUR:    DS   2           ;Sauv coord curseur
410          ;
411          ;
412          END

```

Si vous ne désirez pas entrer les nombreuses lignes de code de ce programme, vous pouvez entrer le chargeur Basic suivant :

```
1000 REM -----
1010 REM Chargeur Basic du traitement de texte
1020 REM -----
1030 FOR I=&B000 TO &B27F
1040   READ A$
1050   A$="&"+A$
1060   POKE I,VAL(A$)
1070 NEXT I
1080 CALL &B000
1090 END
1100 REM -----
2000 DATA 3E,2,CD,E,BC,21,0,90,22,80,82,21,0,90,1,0
2010 DATA 10,16,20,72,B,23,7B,B1,20,F9,CD,8A,BB,CD,6,BB
2020 DATA FE,D,CA,55,80,FE,F0,CA,8A,80,FE,F1,CA,DE,80,FE
2030 DATA F2,CA,11,81,FE,F3,CA,2B,81,FE,1A,CA,9D,81,FE,17
2040 DATA CA,37,81,FE,10,CA,EB,81,FE,7F,CA,5,82,FE,11,C8
2050 DATA C3,2F,82,18,C5,CD,70,82,2A,80,82,11,50,0,19,22
2060 DATA 80,82,CD,7B,BB,7D,FE,19,2B,C,CD,7B,BB,26,1,2C
2070 DATA CD,75,BB,C3,1A,80,6,1,AF,CD,4D,BC,CD,52,82,CD
2080 DATA 7B,BB,26,1,CD,75,BB,C3,1A,80,CD,70,82,CD,7B,BB
2090 DATA 7D,FE,1,2B,16,2A,80,82,11,50,0,ED,52,22,80,82
2100 DATA CD,7B,BB,2D,25,CD,75,BB,C3,1A,80,21,0,90,ED,5B
2110 DATA 80,82,ED,52,20,A,CD,7B,BB,25,CD,75,BB,C3,1A,80
2120 DATA 6,0,AF,CD,4D,BC,2A,80,82,11,50,0,ED,52,22,80
2130 DATA 82,CD,52,82,CD,7B,BB,25,CD,75,BB,C3,1A,80,CD,70
2140 DATA 82,2A,80,82,11,50,0,19,22,80,82,CD,7B,BB,7D,FE
2150 DATA 19,2B,B,CD,7B,BB,25,2C,CD,75,BB,C3,1A,80,6,1
2160 DATA AF,CD,4D,BC,CD,52,82,CD,7B,BB,25,CD,75,BB,C3,1A
2170 DATA 80,CD,7B,BB,7C,FE,1,CA,1A,80,CD,70,82,CD,7B,BB
```

```

2180 DATA 25,25,CD,75,BB,C3,1A,80,CD,7B,BB,7C,FE,50,CA,1A
2190 DATA 80,CD,70,82,C3,1A,80,CD,7B,BB,22,87,82,45,2A,80
2200 DATA 82,11,50,0,37,3F,ED,52,10,FC,19,22,83,82,11,0
2210 DATA 90,ED,52,7C,B5,CA,1A,80,CD,7B,BB,6,0,4C,2A,80
2220 DATA 82,9,2B,7E,CD,5A,BB,6,0,AF,CD,4D,BC,21,1,1
2230 DATA CD,75,BB,2A,80,82,22,85,82,2A,83,82,11,50,0,ED
2240 DATA 52,22,80,82,CD,52,82,2A,87,82,CD,75,BB,2A,85,82
2250 DATA 11,50,0,37,3F,ED,52,22,80,82,C3,1A,80,CD,7B,BB
2260 DATA 22,87,82,6,0,4C,2A,80,82,9,2B,7E,CD,5A,BB,6
2270 DATA 1,AF,CD,4D,BC,21,19,1,CD,75,BB,2A,80,82,22,85
2280 DATA 82,ED,5B,87,82,3E,1A,93,47,11,50,0,19,10,FD,22
2290 DATA 80,82,CD,52,82,2A,85,82,11,50,0,19,22,80,82,2A
2300 DATA 87,82,CD,75,BB,C3,1A,80,CD,7B,BB,3E,50,94,47,4
2310 DATA 16,0,5C,2A,80,82,19,2B,54,5D,23,ED,A0,10,FC,CD
2320 DATA 52,82,C3,1A,80,CD,7B,BB,7C,FE,1,CA,1A,80,3E,50
2330 DATA 94,47,4,16,0,5C,2A,80,82,19,2B,54,5D,1B,ED,A0
2340 DATA 10,FC,CD,52,82,CD,7B,BB,25,CD,75,BB,C3,1A,80,FE
2350 DATA 20,DA,1A,80,32,82,82,CD,7B,BB,4C,7C,FE,50,CA,1A
2360 DATA 80,6,0,2A,80,82,9,2B,3A,82,82,77,CD,5A,BB,C3
2370 DATA 1A,80,CD,7B,BB,22,83,82,26,1,CD,75,BB,6,50,2A
2380 DATA 80,82,7E,CD,5A,BB,23,10,F9,2A,83,82,CD,75,BB,C9
2390 DATA CD,7B,BB,4C,6,0,2A,80,82,9,2B,7E,CD,5A,BB,C9
    
```

Assurez-vous que les données entrées sont correctes grâce au programme de checksum qui doit donner le résultat suivant :

**62 CE 8C D0 71 DD 24 5C 7B AF AC F1 FE E7 CD 4 2E 27 5A BD  
F9 67 C9 D5 7F 9D 48 97 B3 A1 D7 22 A5 1F 33 CB 46 6B 8B E1**

Si une des données affichées par le programme de checksum n'est pas identique à celles indiquées ci-dessus (supposons qu'il s'agisse de la donnée *n*), la nième ligne de données contient très certainement une ou plusieurs fautes de frappe...

